

## Correction en fréquence des amplificateurs opérationnels. Exemples.

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :  
[page d'accueil](#)

d'autres pdf, sur différents sujets :  
[liste des PDF](#)

Les A. Op. performants sur le plan dynamique peuvent osciller sur des montages simples. Ce document illustre cette situation au travers d'exemples concrets.


### 1. Étude, prédétermination théorique

#### 1.a Position du problème

On désire réaliser un inverseur à base d'un amplificateur opérationnel. Le schéma, traditionnel, est donné ci-contre.

On fixe  $R_a = R_b = 1 \text{ k}\Omega$ .

Par suite,  $R = 470 \Omega$ . (Son rôle n'intervient pas dans la fonction inversion. Cette résistance d'équilibrage est placée là pour annuler le défaut dû à  $I_{BIAS}$  [Réf 1]).



Schéma

Le montage peut être étudié avec le formalisme des systèmes bouclés :

•  $A(j\omega)$  représente la fonction d'amplification de l'amplificateur opérationnel seul.

•  $a(j\omega)$  est l'atténuation d'entrée =  $-\frac{R_a}{R_a+R_b}$ . [Réf 2]

•  $b(j\omega)$  est l'atténuation de retour =  $\frac{R_b}{R_a+R_b}$  [Réf 2]




Schéma bloc  
correspondant

Rappelons que la FTBF( $j\omega$ ) d'un tel système est :  $\frac{v_s(j\omega)}{v_e(j\omega)} = a \frac{A(j\omega)}{1 + A(j\omega)b}$

Dans le cas d'un amplificateur parfait ( $A \rightarrow \infty$ ), :  $\frac{v_s}{v_e} = \frac{a}{b} = -\frac{R_a}{R_b}$ . Avec  $R_a = R_b$ , on a donc un inverseur.

Dans le cas d'un amplificateur réel, ( $A$  fonction de la fréquence), il faut tout d'abord vérifier les conditions de stabilité. Si le système bouclé est stable, alors on retrouvera un fonctionnement en inverseur dans une gamme de fréquence où  $A(j\omega)b \gg 1$ .

La recherche des conditions de stabilité peut se faire par une analyse de la FTBO :  $A(j\omega) b$


La « data sheet » de l'amplificateur opérationnel utilisé donne  $A(j\omega)$ , ou  $A(jf)$ ,

Le montage donne  $b = \frac{R_b}{R_a+R_b} = 0,5$ .

L'étude de ce simple circuit doit donc comporter une prédétermination de la stabilité, avec recherche d'un système de correction si nécessaire. C'est ce que nous proposons de faire.


## 1.b Prédétermination de la stabilité

On choisit d'utiliser un amplificateur opérationnel rapide, le LM 308, de la famille LM 108, LM 208. Un extrait de ses caractéristiques, données par le constructeur (source National Semiconductor), donne :



**Réponse harmonique du LM 308 selon les valeurs de Cf ou Cs, condensateurs de compensation.**

Source : <http://www.national.com/pf/LM/LM308.html>



Explications :


- Cf est le condensateur à placer par l'utilisateur entre les broches 1 et 8 du boîtier. Ce condensateur permet de faire la compensation « 1 pôle ». [Réf. 3]. Les valeurs indiquées sont 3 pF et 30 pF. Ces valeurs sont usuelles pour ce type de compensation, sur de nombreux amplificateurs opérationnels. Il est précisé ici la valeur recommandée :  $C_f > C_o R_1 / (R_1 + R_2)$ , avec  $C_o = 30 \text{ pF}$ .

Dans notre application, cela donne  $C_f > 15 \text{ pF}$ .

- Dans l'autre schéma, il s'agit d'une variante de la compensation 1 pôle. Cs est le condensateur à placer par l'utilisateur entre les broches 8 du boîtier et la masse. La valeur indiquée est de 100 pF.

Le constructeur ne donne pas d'indication sur la réponse en fréquence du LM 308 sans aucun condensateur de compensation. Une possibilité est d'extraire, par extrapolation, une réponse idéalisée du LM 308. Il suffit à partir de la réponse à  $C_f = 30 \text{ pF}$ , et de la réponse à  $C_f = 3 \text{ pF}$ , de positionner la réponse à  $C_f = 0$ .

Remarque : le cas de figure  $C_f = 0$  n'existe pas en pratique. Un condensateur parasite est toujours présent entre les broches 1 et 8 du circuit : les isolants placés entre les différents conducteurs (broches du circuit intégré, supports, pistes, plaque..) peuvent former un condensateur de quelques 0,1 pF. Cette valeur peut être beaucoup plus faible en CMS.



**Réponse harmonique de l'amplificateur opérationnel LM 308 seul.**  
**Ce tracé est asymptotique et approximatif (voir texte)**

C'est la réponse d'un amplificateur de « type 2 » [Réf 4].

Le plateau en basse fréquence est à 109,5 dB (valeur spécifiée par le constructeur).


La première fréquence de cassure est à 50 Hz (valeur estimée à  $C_f$  quasi nul),

La deuxième fréquence de cassure est à 1,5 MHz (valeur estimée à  $C_f$  quasi nul).

On idéalise ainsi cette réponse à celle d'un deuxième ordre, ce qui est une hypothèse de travail suffisante pour les calculs qui vont suivre. Expérimentalement, la réponse est plus complexe au-delà du MHz.

Pour étudier la stabilité d'un système bouclé, une méthode consiste à tracer les diagrammes de Bode de la fonction de transfert en boucle ouverte, soit  $A(jf).b$ , et de déduire la marge de phase

Dans le cas d'un montage inverseur,  $b = 0,5$ . Le tracé de la TBO avec ce LM 308 modélisé est donc :



**Réponse harmonique de la FTBO, soit  $A(jf).b$ .**  
**Ce tracé est asymptotique et approximatif.**

La courbe de module est simplement obtenue par une translation de 6 dB vers le bas ( $20 \log 0,5$ ) de la réponse du LM 308 seul.

La courbe de phase est inchangée (le terme  $b$  est indépendant de la fréquence).

Sans avoir à tracer les réponses exactes, les réponses asymptotiques montrent, au passage à 0 dB, une phase voisine de  $-180^\circ$ , donc une marge de phase nettement inférieure à  $45^\circ$  : le système ne présente pas des conditions de stabilité satisfaisantes. Si mathématiquement le système n'est pas instable, en pratique, il est possible qu'à cause d'un élément non pris en compte, que le système oscille. Cet élément peut être une charge capacitive sur l'amplificateur opérationnel (apportée par l'équipement de mesure), ou une courbe de phase différente que celle modélisée ici.

Aussi, il est préférable de prévoir un système de correction en fréquence de la fonction  $A(jf)$ .

### 1.c Ajout d'un système de correction

On se propose de mettre en œuvre, en série avec l'amplificateur opérationnel, le circuit de compensation "retard de phase", dont un schéma type est donné ci-contre :

Ce circuit admet comme transmittance complexe :

$$C(j\omega) = \frac{\varepsilon(j\omega)}{\varepsilon'(j\omega)} = \frac{1 + \tau_1 j\omega}{1 + \tau_2 j\omega} \quad \text{avec} \quad \tau_1 = R_1 C \quad \tau_2 = (2R_2 + R_1) C.$$





Schéma du correcteur.

On peut représenter sa réponse harmonique, sous forme de tracé de Bode, soit :

$20 \log |C(j\omega)|$   
 et  $\arg C(j\omega)$

On montre ci-contre les asymptotes, donné ici sur un axe des fréquences :



Réponse en fréquence du correcteur


Nous proposons ici, comme synthèse de réseau correcteur, une méthode rapide et graphique sur la réponse harmonique de l'amplificateur opérationnel seul : on cherche à ramener sa réponse à celle d'un type 1 [Réf 4]. Cette façon de faire est maximaliste : en ayant une telle réponse harmonique de l'amplificateur corrigé, on s'assure d'une marge de phase de  $45^\circ$  pour le cas d'un suiveur (qui est le cas où  $b = 1$ , c'est à dire que la TBO du système est ramenée à  $A(jf)$ ).

Pour ce faire, on cherche que la fréquence de cassure 1,5 MHz passe à 0 dB, et que la réponse harmonique soit formée d'un plateau (obligatoirement inchangée, soit 109,5 dB), et d'une fréquence de coupure basse.

Ces conditions amènent aux valeurs numériques suivantes :  $f_1 = \frac{1}{2\pi\tau_1} = 50\text{Hz}$  et  $f_2 = \frac{1}{2\pi\tau_2} = 5\text{Hz}$

c'est à dire  $\tau_1 = 3,18 \text{ ms}$  et  $\tau_2 = 31,8 \text{ ms}$ .

La réponse harmonique de l'association correcteur + LM308 est alors :



**Réponse harmonique de l'association LM308 + réseau correcteur à retard de phase :**

**Le passage à 0 dB correspond à une phase de  $-135^\circ$  (sur asymptote du module), ce qui est une condition pour le cas du suiveur.**

**Dans notre application (inverseur, donc  $b = 0,5$ ) il faut translater la courbe de gain de  $-6$  dB.**


Dans notre cas d'inverseur, il est facile de repérer où se trouve le passage à 0 dB après la translation vers le bas de 6 dB. On vérifie alors que la marge de phase est supérieure à  $45^\circ$ . Rappelons que cette détermination est faite sur la courbe de gain asymptotique. Sur la courbe de gain réelle, le passage à 0 dB est légèrement décalé, et la marge de phase est très faiblement supérieure.

En fixant arbitrairement C à 150 nF, on aboutit à :  
 $R_1 = 21,2 \text{ k}\Omega$  ;  $R_2 = 95,4 \text{ k}\Omega$ .

Les valeurs normalisées sont :  
 $C = 150 \text{ nF}$  ;  $R_1 = 22 \text{ k}\Omega$  ;  $R_2 = 100 \text{ k}\Omega$ .

On remarque que la résistance  $R = 470 \Omega$ , en série avec 100 k $\Omega$ , perd son intérêt et peut être retirée.

Le schéma résultant est donc :



**Schéma de l'inverseur avec l'amplificateur opérationnel équipé de son correcteur retard de phase.**

Ce schéma final peut donc être testé.

Mais, pour conforter cette prédétermination théorique, on peut facilement simuler ce schéma à l'aide de Pspice, version d'évaluation. C'est ce que nous nous proposons de faire.

## 2. Simulations PSpice

### 2.a Réponse avec le LM 308 non compensé

Il existe (notamment sur le Web) plusieurs fichiers modèles de LM 308. Il est donc prudent de réaliser plusieurs simulations afin de valider le modèle.

Pour ce faire, on propose de placer le LM 308 en inverseur, et de réaliser plusieurs analyses.

Le fichier.cir est le suivant :

Inverseur a LM308

\* Simule un ampli a base de 308

\* fichier invlm308.cir

\*

```
.subckt LM308 1 2 3 4 5 6 7
```

```
c1 11 12 5.460E-12
```

```
dc 5 53 dx
```

```
de 54 5 dx
```

```
d1p 90 91 dx
```

```
d1n 92 90 dx
```

```
dp 4 3 dx
```

```
egnd 99 0 poly(2) (3,0) (4,0) 0 .5 .5
```

```
fb 7 99 poly(5) vb vc ve vlp vln 0 83.87E6 -80E6 80E6 80E6 -80E6
```

```
ga 6 0 11 12 150.8E-6
```

```
gcm 0 6 10 99 1.508E-9
```

```
iee 10 4 dc 18.00E-6
```

```
hlim 90 0 vlim 1K
```

```
q1 11 2 13 qx
```

```
q2 12 1 14 qx
```

```
r2 6 9 100.0E3
```

```
rc1 3 11 6.631E3
```

```
rc2 3 12 6.631E3
```

```
re1 13 10 3.756E3
```

```
re2 14 10 3.756E3
```

```
ree 10 99 11.11E6
```

```
ro1 8 5 50
```

```
ro2 7 99 25
```

```
rp 3 4 106.4E3
```

```
vb 9 0 dc 0
```

```
vc 3 53 dc 1
```

```
ve 54 4 dc 1
```

```
vlim 7 8 dc 0
```

```
vlp 91 0 dc 6
```

```
vln 0 92 dc 6
```

```
.model dx D(Is=800.0E-18 Rs=1)
```

```
.model qx NPN(Is=800.0E-18 Bf=6.000E3)
```

```
.ends
```

\*

\* circuit simule :

```
Vplus 8 0 DC 12 ; pour le lm308
```

```
Vmoins 9 0 DC -12 ; pour le lm308
```

```
Vin 1 0 pulse (-0.2 +0.2 -0.05u 0.1u 0.1u 19.9u 40u) ac 1
```

```
X1 4 2 8 9 3 5 6 lm308 ; amplificateur e+ e- V+ V- S Comp Comp
```

```
Cc 5 6 0.3p ; compensation par défaut
```

```
Rb 1 2 1k ; entree
```

```
Ra 2 3 1k ; contre reaction
```

```
R 4 0 470 ; equilibrage
```

```
CL 3 0 120p ; charge
```

```
RL 3 0 1Meg
```


```
.AC dec 100 0.1 10meg ; pour essai harmonique
```

```
.TRAN .01u 50u 0u 0.01u ; pour réponse au signal carre
```

```
.probe
```

```
.END
```

#### Schéma correspondant :



$$C_c = 0,3 \text{ pF} ; C_L = 120 \text{ pF}$$

- Les numéros représentent les numéros de nœud du fichier invlm308.cir ci-contre (et non pas les n° de broches du circuit intégré).

- La description de LM 308 est réalisée par le sous-circuit (directive .subckt).

- Le signal d'entrée appliqué en V(1) est :

- soit une source sinusoïdale d'amplitude unitaire, et de fréquence variable, utilisée uniquement pour l'analyse AC

- soit une source représentant un signal carré (VPULSE), utilisé uniquement pour l'analyse TRAN

- L'amplificateur est chargé par un condensateur  $C_L = 120 \text{ pF}$  en parallèle avec  $1 \text{ M}\Omega$ , qui représente le câble coaxial relié à l'oscilloscope.


- Les accès au condensateur de compensation (réservé à la compensation 1 pôle) ne doivent pas être flottantes en simulation. Il est nécessaire de placer un condensateur, même minime, pour que le modèle soit représentatif. Il a été mis ici un condensateur de  $0,3 \text{ pF}$ . Cette valeur peut représenter un condensateur parasite sur le circuit. C'est équivalent à l'absence de compensation 1 pôle. (Une compensation 1 pôle se ferait avec  $C_c = 3 \text{ pF}$  à  $30 \text{ pF}$ ).

- La tension différentielle  $\varepsilon$  est  $V(4)-V(2)$ . L'amplification de la chaîne directe est donc  $V(3)/(V(4)-V(2))$ .



### 2.a.1 Analyse harmonique (directive .AC)

Cette analyse AC permet d'avoir le comportement fréquentielle du LM 308 :




**Réponse harmonique obtenue après l'analyse AC de Pspice.**

Ce modèle de LM 308 montre une allure de deuxième ordre avec un gain statique de 109,3 dB, une fréquence de coupure à 260 Hz, et une deuxième cassure à 2 MHz où le gain est à encore 28,8 dB. Au-delà, la réponse en fréquence simulée est probablement moins fiable : il est peu probable de disposer d'une fréquence de transition supérieure à 10 MHz pour ce type de circuit. Cette simulation montre un modèle de LM 308 plus rapide que la réalité. Nous retrouverons cette remarque par la suite.

Néanmoins, ce tracé montre bien que le montage inverseur sera instable : en traduisant la réponse du LM 308 de - 6 dB (20 log 0,5), un zoom montre que : le gain vaut 2,4 dB au passage à -180 °, et ce à la fréquence 7,8 MHz , la phase vaut -183° au passage à 0 dB : la marge de phase est négative.

### 2.a.2 Analyse temporelle (directive .TRAN)

Une analyse temporelle, ou transitoire, permet de vérifier le comportement de ce montage :



Le signal d'entrée carré présente des niveaux haut et bas de 200 mV, et une fréquence de 25 kHz (période de 40 us).

Le signal de sortie devrait être ce même carré inversé. On observe la superposition d'une sinusoïde.

**Réponse temporelle obtenue après l'analyse .TRAN de Pspice.**

Un zoom permet de mesurer la période des oscillations : 132 ns, soit une fréquence d'oscillation de 7,6 MHz. Ce relevé est cohérent avec la réponse harmonique précédente.


## 2.b Réponse simulée avec circuit de compensation

On place en série avec les entrées différentielles de l'amplificateur opérationnel le réseau correcteur. Le sous-circuit est inchangé. Il n'est pas reporté dans le listing ci-après.

Le nouveau fichier.cir montre :

```
* circuit simule :
Vplus 8 0 DC 12 ; pour le lm308
Vmoins 9 0 DC -12 ; pour le lm308
Vin 1 0 pulse (-0.2 +0.2 -0.05u 0.1u 0.1u 19.9u 40u) ac 1
X1 4 2 8 9 3 5 6 lm308 ; amplificateur e+ e- V+ V- S Comp Comp
Cc 5 6 0.3p ; compensation par défaut
Rb 1 20 1k ; entree
Ra 20 3 1k ; contre reaction
CL 3 0 120p ; charge
RL 3 0 1Meg ; charge
R2a 20 2 100k
R1 2 7 22k
C 7 4 150n
R2b 4 0 100k
.AC dec 100 0.1 10meg ; pour essai harmonique
.TRAN .01u 50u 0u 0.01u ; pour réponse au signal carre
.probe
.END
```

Schéma correspondant :




$C_c = 0,3 \text{ pF}$  ;  $C_L = 120 \text{ pF}$   
 $R_{2a} = R_{2b} = 100 \text{ k}\Omega$  ;  $R_1 = 22 \text{ k}\Omega$   
 $C = 150 \text{ nF}$

*Le réseau de compensation retard de phase est placé. Cela ajoute 2 numéros de nœud supplémentaires, le 20 et le 7.*

**2.b.1 Analyse harmonique (directive .AC)**

La même analyse AC permet de vérifier les conditions de stabilité



*Réponse harmonique obtenue après l'analyse AC de Pspice.*

Comme le montrait l'étude théorique, cette compensation est maximaliste : sur la réponse de l'amplificateur opérationnel corrigé, on observe :

9° de marge de phase (phase à -171 ° lors du passage à 0 dB)

4,6 dB de marge de gain positive (gain à -4,6 dB lors du passage à - 180 °).

Ce qui signifie qu'un montage suiveur sera stable (mais avec des marges insuffisantes).

La même courbe de gain translatée de - 6 dB montre que le montage inverseur présentera, quant à lui :

10,6 dB de marge de gain,

23 ° de marge de phase.

Le système inverseur sera stable, mais, à cause de marge encore insuffisante, présentera néanmoins un fort rebond lors de la réponse indicielle.


C'est ce que nous nous proposons de visualiser, en 2 essais :

essai petits signaux, pour rester dans le domaine linéaire,

essai plus grands signaux, pour observer des éventuelles distorsions.

## 2.b.2 Analyse temporelle (directive .TRAN)


### Petits signaux :



Essai avec un signal carré d'amplitude 0,2 V en entrée :

Le montage inverseur fonctionne bien, mais présente un net dépassement lors de la transition. Un zoom permet de chiffrer ce phénomène transitoire :

### Réponse temporelle obtenue après l'analyse .TRAN de Pspice.




A cette échelle, on peut mesurer :

un temps de réponse de 0,5  $\mu$ s,  
un premier dépassement de 30 %.

Ce dernier chiffre montre que le correcteur est encore mal réglé.

### Agrandissement de la réponse temporelle précédente


### Grands signaux :



Vérifions le comportement en plus grande amplitude : le signal carré d'entrée a une amplitude de 5 V.

La simulation montre que le système n'est pas linéaire.

### Réponse indicielle en grands signaux



La forme du signal montre que l'on n'est pas en limitation dû au slew rate. La sortie croît d'environ 10 V en 0,5  $\mu$ s, soit 20 V /  $\mu$ s, ce qui, comme annoncé, montre que ce modèle de LM 308 donne une réponse dynamique optimiste.

### Agrandissement de la réponse temporelle précédente

En conclusion, bien qu'avec un modèle du LM 308 sujet à critiques, les simulations ont confirmé que le réseau retard de phase permettait de stabiliser le montage inverseur. On peut passer à l'expérimentation.




### 3. Essais expérimentaux

#### 3.a Montage sans circuit de compensation

Dans un premier temps, on câble le montage SANS circuit de compensation : on sait que le montage est instable. (En général, il est interdit de mettre en route un système bouclé dont on n'est pas sûr de sa stabilité. Mais ici, cet essai est sans risque).

On choisit de placer en entrée un signal carré **petits signaux** (200 mV, 25 kHz).



*Trace supérieure : signal d'entrée (200 mV / c)*  
*Trace inférieure : signal de sortie (200 mV / c)*


Le système oscille, (même sans injecter de signal d'entrée) mais on "voit" néanmoins que la fonction inversion est réalisée : le signal de sortie présente un carré de 200 mV d'amplitude, sur laquelle se superpose une oscillation de 860 kHz, d'amplitude voisine de 200 mV.

On peut même remarquer que le signal d'entrée est perturbé par l'oscillation générée par l'amplificateur opérationnel.

#### 3.b Système corrigé par retard de phase

Dans un deuxième temps, on insère la cellule retard de phase, en amont de l'amplificateur opérationnel.

On reste dans le domaine **petits signaux**. On conserve le même essai, pour une meilleure comparaison.




*Trace supérieure : signal d'entrée (200 mV / c)*  
*Trace inférieure : signal de sortie (200 mV / c)*

Le système présente un régime oscillatoire amorti (un zoom permet de mesurer un premier dépassement de 25%), et un  $t_{r5\%} = 4\mu s$ .

L'inversion est néanmoins réalisée (ici un offset dû à l'amplificateur opérationnel est présent). Il serait possible d'améliorer cette réponse à l'échelon par un choix d'autres valeurs numériques des composants formant le réseau à retard de phase.

On passe en **grands signaux** : l'amplitude de l'entrée est de 5 V.



*Trace supérieure : signal d'entrée (5 V / c)*  
*Trace inférieure : signal de sortie (5 V / c)*

Le signal de sortie est dégradé : le temps de réponse à 5% dépasse les 10  $\mu s$ .


Dans la partie quasi-linéaire de cette réponse, on peut mesurer la pente : 8 V / 5  $\mu s$ , soit 1,6 V /  $\mu s$ .

On n'est pas encore dans la limitation du slew rate

En première conclusion, le réseau correcteur proposé peut être jugé satisfaisant. Il est néanmoins intéressant de réaliser d'autres corrections, afin de chercher une amélioration.

### 3.c Système corrigé par compensation 1 pôle


Faisons la comparaison avec une compensation 1 pôle, sans réseau à retard de phase. La data sheet préconise un condensateur de 15 pF. Faisons déjà un essai avec 10 pF.



*Trace supérieure : signal d'entrée (200 mV / c)  
Trace inférieure : signal de sortie (200 mV / c)*

On refait l'essai en **petits signaux** :

Avec  $C = 10$  pF, la réponse petits signaux est améliorée : moins de dépassement, et temps de réponse un peu inférieur à  $2 \mu\text{s}$ .



*Trace supérieure : signal d'entrée (5 V / c)  
Trace inférieure : signal de sortie (5 V / c)*

Et en **grands signaux** :

L'inverseur doit passer de  $-5$  V à  $+5$  V : il lui faut plus de  $25 \mu\text{s}$  (valeur moyenne).

La valeur  $10$  pF du condensateur de compensation limite la montée du signal de sortie à une pente de  $0,4$  V /  $\mu\text{s}$ . Une valeur supérieure aurait dégradé encore plus les variations de tension de sortie.


En conclusion de ces 2 compensations :

- en grand signaux, sur le strict plan du temps de réponse, on privilégiera le réseau à retard de phase,
- en petits signaux, les 2 compensations peuvent être sensiblement identiques selon les valeurs numériques des composants.

Il peut être judicieux de trouver un compromis, comme par exemple l'association des 2 principes de correction.

### 3.d Système corrigé par retard de phase et compensation 1 pôle

Il s'agit là d'un essai intuitif : on remplace le précédent correcteur retard de phase, et on ajoute une compensation un pôle avec un condensateur de très faible valeur.




*Trace supérieure : signal d'entrée (200 mV / c)  
Trace inférieure : signal de sortie (200 mV / c)*

**Petits signaux** :

La présence de  $C = 1,5$  pF (ce qui est faible) en plus du réseau correcteur permet d'améliorer la réponse indicielle.


voir zoom ci après.





*Trace supérieure : signal de sortie (75 mV / c)  
Trace inférieure : rappel du signal d'entrée (200 mV / c)*

Un zoom permet d'apprécier la justesse de  $C = 1,5 \text{ pF}$  : le temps de réponse à 5 % est à  $2 \mu\text{s}$ , sans dépassement marqué.



*Trace supérieure : signal d'entrée (5 V / c)  
Trace inférieure : signal de sortie (5 V / c)*

En **grands signaux**, on n'est pas limité par le phénomène de slew rate : la compensation 1 pôle est faite avec un condensateur de très faible valeur.

Le temps de réponse à 5 % dépasse encore les  $10 \mu\text{s}$ .

## 4. CONCLUSION

Sur un exemple a priori simple, nous avons exploité des bases d'électronique et d'automatique.

- En modélisant l'amplificateur opérationnel par un système du deuxième ordre, l'étude théorique a été simplifiée, et la détermination graphique du correcteur facilitée.
- Les simulations ont permis de conforter l'étude. Il faut reconnaître, que pour des essais plus pointus, il aurait fallu un modèle plus fiable d'amplificateur opérationnel. Mais la conclusion sur la stabilisation avec le correcteur à retard de phase a été confirmée.
- Les essais ont montré que le réseau correcteur de phase prédéterminé permettait de stabiliser le système. Néanmoins, les réponses indicielles montraient un dépassement, ce qui signifiait que l'on pouvait améliorer le correcteur. Un autre essai a montré que la correction 1 pôle assurait également la stabilité, mais au détriment d'une dégradation de la réponse dynamique en grands signaux. D'où la solution finale, obtenue par une double correction.

Les références font appel au livre « **Comprendre l'électronique par la simulation** » :

- [Réf. 1]. page 293
- [Réf. 2]. page 312
- [Réf. 3]. page 319
- [Réf. 4]. page 316

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)